

## HORIZONTAL SWITCH

### PURPOSE

Specifies whether subsequent plots are drawn in a vertical (the normal direction) or horizontal direction.

### DESCRIPTION

This command is typically used to draw bars, spikes, and box plots down rather than across the page. Many analysts find that it is easier to make visual comparisons down the page rather across it. For this reason, the horizontal format is sometimes preferred, especially when used in conjunction with a portrait page orientation.

### SYNTAX

HORIZONTAL SWITCH <ON/OFF>

where ON specifies that subsequent plots are drawn horizontally while OFF specifies that they are drawn vertically.

### EXAMPLES

HORIZONTAL SWITCH ON  
HORIZONTAL SWITCH OFF

### NOTE 1

The HORIZONTAL SWITCH command with no arguments sets the horizontal switch to the default.

### NOTE 2

The BAR DIRECTION and SPIKE DIRECTION commands perform a similar function specifically for bars and spikes. The HORIZONTAL SWITCH command works for all plot types (although the plot characters may need to be set to more appropriate choices). This command automatically reverses the x and y coordinates for the plot (the BAR DIRECTION and SPIKE DIRECTION commands do not).

### DEFAULT

The default is off (vertical direction).

### SYNONYMS

None

### RELATED COMMANDS

BAR DIRECTION	=	Specify the direction of plot bars.
SPIKE DIRECTION	=	Specify the direction of plot spikes.
BAR	=	Sets the on/off switches for plot bars.
SPIKE	=	Sets the on/off switches for plot spikes.
BOX PLOT	=	Generate a box plot.
ORIENTATION	=	Sets the page orientation.

### APPLICATIONS

Presentation Graphics

### IMPLEMENTATION DATE

Pre-1987

PROGRAM

```

LET Y = DATA 2 5 9 15 28; LET X = DATA 81 82 83 84 85
X1LIMITS 81 85; XTIC OFFSET 1 1
X1LABEL YEAR; MINOR X1TIC MARK NUMBER 0
Y1LABEL SALES (IN MILLIONS OF DOLLARS)
MULTIPLY 2 2; MULTIPLY CORNER COORDINATES 0 0 100 100
.
TITLE VERTICAL BAR CHART; BAR ON
LINE BLANK; BAR WIDTH 0.5
PLOT Y VS X
.
HORIZONTAL SWITCH ON
TITLE HORIZONTAL BAR CHART
YLIMITS 81 85; YTIC OFFSET 1 1
MINOR YTIC MARK NUMBER 0; Y1LABEL YEAR
X1LABEL SALES; XLIMITS ; XTIC OFFSET 0 0; MINOR XTIC MARK NUMBER DEFAULT
PLOT Y VS X
.
TITLE HORIZONTAL SPIKE CHART; SPIKE ON; BAR OFF
PLOT Y VS X
.
TITLE HORIZONTAL BOX PLOT
SPIKE OFF; YLIMITS; X1LABEL; Y1LABEL; YTIC OFFSET 0 0
LET X = SEQUENCE 0 1 999; LET X = X/100; LET X = INT(X) + 1
LET Y = NORMAL RANDOM NUMBERS FOR I = 1 1 1000
FENCES ON; LINES BOX PLOT; CHARACTER BOX PLOT
CHARACTER SIZE 0.9 0.9 0.9 0.9 0.9 0.9 0.9
CHARACTER VBAR BLANK BLANK VBAR BLANK BLANK VBAR
BOX PLOT Y X
END OF MULTIPLY
    
```

